```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

The next examples show that trivial reduction techniques can also be used to simplify the presentation of classical linear differential time-delay systems.

We first start with a wind tunnel model studied in A. Manitius, "Feedback controllers for a wind tunnel model involving a delay: analytical design and numerical simulations", IEEE Trans. Autom. Contr., 29 (1984), pp. 1058-1068.

Let us define the commutative polynomial ring $A = \mathbb{Q}(\zeta,k,a,\omega)[d,\delta]$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d,t],dual_shift=[delta,s],polynom=[t,s],
> comm=[zeta,k,a,omega],shift_action=[delta,t,h]):
```

where d $y(t) = \frac{d}{dt}\, y(t)$ is the differential operator and $\delta\, y(t) = y(t-h)$ is the time-delay operator.

Let us also consider the matrix P defining the wind tunnel model

```
> P := matrix(3,4,[d+a,k*a*delta,0,0,0,d,-1,0,0,omega^2,d+2*zeta*omega,-omega^2]);
```

$$P := \begin{bmatrix} d+a & k\,a\,\delta & 0 & 0 \\ 0 & d & -1 & 0 \\ 0 & \omega^2 & d+2\,\zeta\,\omega & -\omega^2 \end{bmatrix}$$

and the A-module L finitely presented by P. We can use Stafford's reduction to try to simplify the presentation of L by removing trivial unimodular elements, namely, unimodular elements which are A-linear combinations of the generators of L but not $A_2(Q(\zeta,k,a,\omega))$-linear ones.

```
> S := StaffordReduction(P,A,"reduce_relations"=true);
```

$$S := \left[\begin{bmatrix} d+a & -\frac{k\,a\,d\,\delta}{\omega^2} - \frac{2\,k\,a\,\zeta\,\delta}{\omega} & k\,a\,\delta \\ 0 & \frac{d^2}{\omega^2} + \frac{2\,\zeta\,d}{\omega} + 1 & -d \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \right]$$

We get that L is isomorphic to the A-module L_1 finitely presented by S[1], where the A-isomorphism γ from L_1 onto L is induced by S[2].

```
> TestIso(S[1],P,S[2],A);
```

$$\mathit{true}$$

We can try again to find a trivial reduction of the presentation of L_1.

```
> T := StaffordReduction(S[1],A,"reduce_relations"=true);
```

$$T := [\begin{bmatrix} d+a & -\frac{k\,a\,\delta}{\omega^2} \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & d+2\,\zeta\,\omega & -\omega^2 \end{bmatrix}]$$

Thus, L_1 is isomorphic to the A-module L_2 finitely presented by T[1], where the A-isomorphism β from L_2 onto L_1 is induced by T[2].

```
> TestIso(T[1],S[1],T[2],A);
```

$$true$$

If we define U = T[2] S[2], namely,

```
> U := Mult(T[2],S[2],A);
```

$$U := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & d + 2\,\zeta\,\omega & -\omega^2 \end{bmatrix}$$

then $\theta = \gamma$ o β is induced by U and is an A-isomorphism:

```
> TestIso(T[1],P,U,A);
```

$$true$$

Hence, we get that the linear differential time-delay system P y = 0 defining the wind tunnel model is equivalent to the single differential time-delay equation:

$$\tfrac{d}{dt}\,\mathrm{x}(t) + a\mathrm{x}(t) - \tfrac{k\,a}{\omega^2}\,\mathrm{u}(t-h) = 0.$$

Finally, θ^{-1} is induced by Q[2], where Q[2] is the second entry of Q defined by

```
> Q := InverseMorphism(T[1],P,U,A);
```

$$Q := \left[\begin{bmatrix} 1 & 0 \\ 0 & -\dfrac{1}{\omega^2} \\ 0 & -\dfrac{d}{\omega^2} \\ 0 & -\dfrac{\omega^2 + d^2 + 2\,d\,\zeta\,\omega}{\omega^4} \end{bmatrix}, \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} \right]$$

and we can finally check again that θ^{-1} is an A-isomorphism:

```
> TestIso(P,T[1],Q[1],A);
```

$$true$$